

Digital Voltage Sensitive Relay

1. Systembeschreibung

Zur schnelleren Aufladung von zwei Batteriebänken (Starter und Domestic) wird die nachfolgend beschriebene Schaltung parallel zu dem üblicherweise verwendeten Linearreglern eingesetzt.

Hierfür werden, pro Batteriebank je ein Power Relais mit 70A Schaltleistung verwendet, welche von einem PIC 16F627A Mikrokontroller der Firma Microchip, gesteuert werden. Der selber Kontroller wird auch zu Steuerung des Eingangsmultiplexers und des AD- Wandlers verwendet.

Priorität bei der Aufladung hat die Starter Batterie welche bis auf 14,25 V aufgeladen werden soll. Nach Beendigung der Aufladung der Domestic Bank ebenfalls auf 14,25V, werden beide VSR Relais abgetrennt, und beide Bänke dem Linearregler überlassen, der dann dafür sorgt, dass beide Bänke auf die Erhaltungsladung von 13,8V gehalten werden.


Nach Abschluss der Schnellaufladung werden die Spannungen beider Bänke über den Kontroller kontinuierlich weiter überwacht. Unterschreitet die Spannung der Starter Batterie den Wert von $U_{smin} < 12,0V$, oder die der Domestic Batterien den Wert von $U_{dmin} < 12,5V$ so wird der Linearbetrieb unterbrochen und das VSR Relais der betroffenen Batterie wieder eingeschaltet, und zwar so lange bis die Spannung der zugeschalteten Batterie wieder auf 13,5V angehoben ist. Danach geht die Schaltung wieder in den Linearbetrieb zurück, und zu der zyklischen Überwachung der mindest Spannungen von 12,0 bzw. 12,5V.

2. Eingangsmultiplexer

Als Eingangsverstärker wird der Norton Verstärker LM3900 wie nachfolgend beschrieben, verwendet. Für das Umschalten zwischen den Eingangssignalen für die Starterbatteriespannung (U_s) und der domestic Batteriespannung (U_d), wird der Analogschalter CD4066BE verwendet.

Da der nachfolgender AD- Wandler mit einer 5V Referenz betrieben wird, und die maximale Eingangsspannung am Norton Verstärker 15V nicht überschreitet, wird er mit eine Versorgungsspannung $10 < V_{cc} < 15V$ betrieben und seine Verstärkung wird auf 1/3 eingestellt.

Dem Datenblatt des Analogschalters kann entnommen werden, dass er im geschlossenen Zustand einen $R_s < 0,5k$ aufweist, und im offenen zustand einen Leckstrom von $I_s < 0,05\mu A$ durchlässt. Damit der Leckstrom das Ergebnis am Ausgang des Verstärkers nicht das Ergebnis verfälscht, werden von einen an 5V angeschlossenen Spannungsteiler 0,05 μA zur Kompensation von I_s am Plus Eingang des Verstärkers eingespeist. Das weglassen dieser Kompensation führt allerdings nicht zu untragbare Fehlmessungen.



Um unter Berücksichtigung von R_{smax} von 0,5k sicher zu stellen, dass der Fehler am kurzgeschlossenen Eingang 0,1% der Maximal anliegenden Spannung von $U_d = 15V$ nicht übersteigt, so wird **$R_1 = 1M$** gewählt. Da schließlich auch damit gerechnet werden muss, dass die zu Schaltende Spannung die 5V übersteigen könnte, so muss der Analogschalter mit V_{cc} betrieben und mit V_{cc} auch angesteuert werden.

Wählt man **$R_2 = 330k$** so erhält man:

$$\frac{U_s - V_{BE}}{R_1 + R_2} - \frac{V_{BE}}{R_2} = \frac{U_a - V_{BE}}{R_4} - \frac{V_{BE}}{R_3} \quad \text{bzw.} \quad U_a = U_s \frac{R_4}{R_1 + R_2} + V_{BE} \left(1 + \frac{R_4}{R_3} - \frac{R_4}{R_2} - \frac{R_4}{R_1 + R_2} \right)$$

Mit $R_4 / (R_1 + R_2) = 1/3$ und $1 + R_4 / R_3 - R_4 / R_2 - R_4 / (R_1 + R_2) = 0$, erhält man:

$$\mathbf{R_4 = (R_1 + R_2) / 3 = 443,33k} \quad \text{und} \quad R_3 = R_2 \frac{R_1 + R_2}{R_1 - R_2} = 655k$$

R_3 wird zusammengesetzt aus: **$R_{31} = 270k$** und **$R_{32} = 390k$**

Die Übertragungsfunktion des Norton Verstärkers lautet demnach:

$$\mathbf{V_N = U_a / U_d = 1 / 3}$$

Da die Verarbeitung von Eingangssignalen $< 0,6V$ im vorliegenden Anwendungsfall nicht vorkommt, wird auf ein „Common mode biasing“ verzichtet. Da die Ausgangspegel ebenfalls oberhalb von 0,6V liegen werden, wird auch auf die Einführung einer Ausgangsdiode verzichtet.

3. Analog Digital Wandler

Eingesetzt wird der 8 bit AD- Wandler TLC 549, welcher wird mit eine Referenzspannung von 5V betrieben wird. Für die Skalierung am Eingang des Microcontroller's erhält man:

$$\mathbf{S = 255 [\text{bit}] * V_N / 5 [V] = 17 [\text{bit} / V]}$$


Da nicht sichergestellt werden kann. Dass die Eingangsspannung des Wandlers während der Messung sich nicht ändert, muss jede Messung in zwei aufeinander folgenden Routinen erfolgen.

4. PIC 16F627A

Der Microcontroller wird mit einem Quarz von 4,069 MHz betrieben, welches zu einen instruction cycle von $0,976563\mu s$ führt. Verwendet man als TMR0 Prescaler die Teilung 1/32, so erhält man ein Interrupt alle **$0,976563 * 256 * 32 = 8msec$** .

Erfolgt die Aufladung der Batterien nur über den Alternator (Solarpanels „Dunkel“), so kann auf Grund der Begrenzung der Ausgangsspannung des Alternator's auf etwa 3,8V, eine Endspannung von 14,25V bei den Batterien nicht erzielt werden. Um ein „Aufhängen“ des Rechners in einen solchen Fall zu vermeiden, werden folgende Vereinbarungen getroffen:

- Nach Power ON wird zunächst die Starter Batterie auf 13,5V aufgeladen und anschließend in Richtung 14,25V weitergeladen. Wird 2 Minuten nach Erreichen der 13,5V Schwelle festgestellt, dass die Spannung der Domestic Batteriebank unterhalb ihres Minimum Wertes von 12,5V liegt, wird der Aufladevorgang unterbrochen, und auf die Domestic Bank umgeschaltet.
- Bei der Aufladung der Domestic Batteriebank wird genau so verfahren, d.h. der Ladevorgang wird nach 2 Minuten nach Erreichen der 13,5V Schwelle unterbrochen, wenn festgestellt wird, dass die Spannung der Starter Batterie unter 12,0V gefallen ist.



4.2 Assembler listing

```

;VSR.asm Voltage sensitive relay
;VSR einschalten bei Usmin < 12,0V und Udmin < 12,5V
;VSR ausschalten bei Usmax bzw. Udmax größer 14,25V
;Mittelschwelle 13,5V
;VSR Rise/Fall time < 8msec
;-4,069MHZ/instruction cycle=0,976563µs
;-Lasaros Goumas, Januar 2014
;*****

```

Digital Voltage Sensitive Relay

```
;Assembler directives
list,P16F627
#include <p16F627A.inc>           ;Controller header file
__CONFIG _CP_OFF & _PWRTE_OFF & _WDT_OFF & _XT_OSC
ERRORLEVEL -302                ;Supress Bank SELECTION MESSAGES
;*****

;Zuordnung der verwendeten Register

count      equ    0x20          ;Register für 20µsec Pause
adres      equ    0x21          ;ADW Ergebnis
bit        equ    0x22          ;Zählregister für 8 ADW Durchläufe
highref    equ    0x23          ;14,25V Referenz
midref     equ    0x24          ;13,5V Referenz
slowref    equ    0x25          ;12,0V Referenz
dlowref    equ    0x26          ;12,5V Referenz
w_copy     equ    0x27          ;saved w register
s_copy     equ    0x28          ;saved STATUS register
achtsoll   equ    0x29          ;2x8msec Vorgabe Register
achtist    equ    0x2A          ;2x8msec delay Register
delay      equ    0x2B          ;2sec Vorgabe Register
test       equ    0x2C          ;2sec ist Register
sechszig   equ    0x2D          ;60sec Vorgabe Register
;*****

;Ziele der Register Operationen

w          equ    0            ;w=Zielregister
f          equ    1            ;f=Zielregister
;*****

;Macro- Definitionen

bank0      macro
            bcf    STATUS, 5
            bcf    STATUS, 6
            endm

bank1      macro
            bsf    STATUS, 5
            endm
            org    0x04
;*****

            org    0x00          ;Programmbeginn bei Adresse 00h
            goto   start
;*****

            org    0x04          ;Interrupt service adresse

;Interrupt service vector

            bcf    INTCON, 7      ;Global interrupt disabled
            movwf w_copy          ;Save w in 0x28
            movf  STATUS, 0       ;STATUS in w
            movwf s_copy          ;Save STATUS in 0x29

;Start interrupt routine

            Decfsz   achtsoll, 1 ;2x8msec Vorgabe um 1 reduzieren
```

```

        goto $+02
        bsf  achtist, 0          ;16msec sind vorbei

        decfsz    delay          ;2sec Vorgabe um 1 reduzieren
        goto $+02
        bsf  test, 0            ;2sec sind vorbei

;Interrupt routine Ende

        bcf  INTCON, 2          ;TMR0 interrupt flag (TOIF) löschen

;Restore context and return to the main programm

        movf  s_copy, 0          ;Hole STATUS zurück in w
        movwf STATUS            ;Original werte zurück in STATUS
        movf  w_copy, 0          ;Original Werte zurück in w
        retfie
;*****

;Subroutine Initialisierung

init  clrf  count
      clrf  adres
      clrf  bit
      clrf  achtsoll
      clrf  achtist
      clrf  delay
      clrf  test
      clrf  sechszig
      movlw .242
      movwf highref
      movlw .230
      movwf midref
      movlw .213
      movwf dlowref
      movlw .204
      movwf slowref
      movlw 0x07
      movwf CMCON              ;törn off comparators CM<2:0>=111
      bank1
      movlw 0x01
      movwf TRISA              ;RA0 Eingang: RA<2:1> Ausgänge
      movlw 0xF0
      movwf TRISB              ;RB<3:0> Ausgänge
      bank0
      clrf  PORTA
      clrf  PORTB

      ;TMR0 voreinstellen und starten
      ;TMR0 overflow alle 0,976563 x 32 x 256 = 8 ms

      bcf  INTCON, 7          ;global interrupt disabled
      bank1
      bcf  OPTION_REG, 5      ;gehe zu bank 1
      bsf  OPTION_REG, 2      ;intern. instruct. cycle clock
      bcf  OPTION_REG, 1
      bcf  OPTION_REG, 0      ;PS2,PS1,PS0="100"/prescaler 1:32
      bcf  OPTION_REG, 3      ;prescaler assigned to TMR0
      bank0
      ;zurück zu bank 0
      return

```

Digital Voltage Sensitive Relay

;Subroutine 20µsec Pause

```
adwpause    movlw .5
            movwf count           ;Register count mit 5 laden
            decfsz    count, 1     ;count um 1 reduzieren: Ist count "0"?
            goto    $-01          ;Nein: Wiederholen
            return
```

;Subroutine 8msec pause

```
vsrpause    clrf    achtist
            movlw .2
            movwf achtsoll
            btfss    achtist, 0    ;Sind 16msec vorbei?
            goto    $-01          ;Nein
            return
```

;Subroutine AD - Wandlung

```
adw         clrf    adres           ;wandlungsergebnis register löschen
            call    adwpause       ;20µs pause
            movlw .8
            movwf bit              ;Zaelregister bit mit 8 laden
            bcf    STATUS, 0       ;Carry bit löschen
            bcf    PORTA, 2        ;Chip select enable/Wandler gibt MSB aus
            bsf    PORTA, 1        ;Clock "High"
            btfsc PORTA, 0        ;Ist RA0 auf 0?
            goto    $+03          ;nein
            bcf    adres, 7        ;Wenn "0" MSB in adres löschen
            goto    $+02
            bsf    adres, 7        ;MSB in adres setzen
            rlf    adres, f        ;Inhalt adres nach links schieben
            bcf    PORTA, 1        ;Clock "Low"
            decfsz    bit, f       ;Ist bit register auf 0?
            goto    $-09          ;nein
            rlf    adres, f        ;letztes Nachschieben
            bsf    PORTA, 2        ;Chip select disable
            return
```

;Subroutine VSR auf Starter nachladen

```
svsr        movlw B'00000101'
            movwf PORTB           ;Starter nachladen
            call    vsrpause       ;16 msec warten
            bcf    INTCON, 7       ;GIE disabled
            call    adw
            call    adw
            bsf    INTCON, 7       ;GIE enabled
            bsf    STATUS, 0       ;Flag C setzen
            movf    adres, 0       ;Starterspannung in W
            subwf    midref, 0
            btfsc    STATUS, 0     ;Ist Us > 13,5V?
            goto    $-08          ;Nein: Starter weiter laden
            clrf    PORTB         ;Ja: Starter OFF
            call    vsrpause       ;16 msec warten
            return
```

;Subroutine VSR auf Domestic nachladen

```

dvsr      movlw B'00001010'
          movwf PORTB           ;Domestic nachladen
          call vsrpause         ;16 msec warten
          bcf  INTCON, 7        ;GIE disabled
          call adw
          call adw
          bsf  INTCON, 7        ;GIE enabled
          bsf  STATUS, 0        ;Flag C setzen
          movf adres, 0         ;Domestic Spannung in W
          subwf midref, 0
          btfsc STATUS, 0       ;Ist Ud > 13,5V?
          goto $-08             ;Nein: Domestic weiter laden
          clrf PORTB           ;Ja: Domestic OFF
          call vsrpause         ;16 msec warten
          return
;*****

```

;Hauptprogramm

```

start     call  init

          movlw B'10100000'
          movwf INTCON          ;GIE&TOIE enabled/TOIF cleared

          movlw B'00000101'     ;Starter ON
          movwf PORTB
          call vsrpause         ;16 msec warten
          bcf  INTCON, 7        ;GIE disabled
          call adw
          call adw
          bsf  INTCON, 7        ;GIE enabled
          bsf  STATUS, 0        ;Flag C setzen
          movf adres, 0         ;Starterspannung in W
          subwf midref, 0
          btfsc STATUS, 0       ;Ist Us > 13,5V?
          goto $-08             ;Nein: Starter weiter laden
          movlw .60
          movwf sechszig
          movlw .250
          movwf delay
          bcf  test, 0
          bcf  INTCON, 7        ;GIE disabled
          call adw
          call adw
          bsf  INTCON, 7        ;GIE enabled
          bsf  STATUS, 0        ;Flag C setzen
          movf adres, 0         ;Starterspannung in W
          subwf highref, 0
          btfsc STATUS, 0       ;Ist Us > 14,25V?
          goto $+02             ;Nein;
          goto $+13             ;Ja: Starter OFF
          btfss test, 0         ;Sind 2sec vorbei?
          goto $-0B             ;Nein: Starter weiter laden
          decfsz sechszig       ;JA: Sind 2Min vorbei?
          goto $-10             ;Nein: 2sec timer nachladen
          movlw B'00001001'
          movwf PORTB          ;Ja: Domestic messen
          bcf  INTCON, 7        ;GIE disabled
          call adw
          call adw
          bsf  INTCON, 7        ;GIE enabled

```


Digital Voltage Sensitive Relay

```
bsf STATUS, 0 ;Flag C setzen
movf adres, 0 ;Starterspannung in W
subwf dlowref, 0
btfsc STATUS, 0 ;Ist Ud > 12,5V?
goto $+04 ;Nein:
movlw B'00000101' ;Ja: Starter messen
movwf PORTB
goto $-20 ;2Min timer nachladen
bcf PORTB, 0 ;Starter OFF
call vsrpause ;16 msec warten

movlw B'00001010'
movwf PORTB ;Domestic ON
call vsrpause ;16 msec warten
bcf INTCON, 7 ;GIE disabled
call adw
call adw
bsf INTCON, 7 ;GIE enabled
bsf STATUS, 0 ;Flag C setzen
movf adres, 0 ;Domestic Spannung in W
subwf midref, 0
btfsc STATUS, 0 ;Ist Ud > 13,5V?
goto $-08 ;Nein: Starter weiter laden
movlw .60
movwf sechszig ;2Min timer setzen
movlw .250
movwf delay ;2sec timer setzen
bcf test, 0
bcf INTCON, 7 ;GIE disabled
call adw
call adw
bsf INTCON, 7 ;GIE enabled
bsf STATUS, 0 ;Flag C setzen
movf adres, 0 ;Domestic Spannung in W
subwf highref, 0
btfsc STATUS, 0 ;Ist Ud > 14,25V?
goto $+02 ;Nein:
goto $+13 ;Ja: Domestic OFF
btfss test, 0 ;Sind 2sec vorbei?
goto $-0B ;Nein: Domestic weiter laden
decfsz sechszig ;JA: Sind 2Min vorbei?
goto $-10 ;Nein: 2sec timer nachladen
movlw B'00000110'
movwf PORTB ;Ja: Starter messen
bcf INTCON, 7 ;GIE disabled
call adw
call adw
bsf INTCON, 7 ;GIE enabled
bsf STATUS, 0 ;Flag C setzen
movf adres, 0 ;Starterspannung in W
subwf slowref, 0
btfsc STATUS, 0 ;Ist Us > 12,0V?
goto $+04 ;Nein: Domestic OFF
movlw B'00001010' ;Ja: Domestic messen
movwf PORTB
goto $-20 ;2Min timer nachladen
bcf PORTB, 1 ;Domestic OFF
call vsrpause ;16 msec warten

slinear movlw B'00000100'
movwf PORTB ;Us messen
```

```

        call vsrpause           ;16 msec warten
        bcf  INTCON, 7          ;GIE disabled
        call adw
        call adw
        bsf  INTCON, 7          ;GIE enabled
        bsf  STATUS, 0          ;Flag C setzen
        movf adres, 0           ;Starterspannung in W
        subwf slowref, 0
        btfsc STATUS, 0         ;Ist Us > 12,0V?
        call svsr               ;Nein: Starter nachladen
        goto dlinear

dlinear  movlw B'00001000'
        movwf PORTB            ;Ud messen
        call vsrpause           ;16 msec warten
        bcf  INTCON, 7          ;GIE disabled
        call adw
        call adw
        bsf  INTCON, 7          ;GIE enabled
        bsf  STATUS, 0          ;Flag C setzen
        movf adres, 0           ;Domestic Spannung in W
        subwf dlowref, 0
        btfsc STATUS, 0         ;Ist Ud > 12,5V?
        call dvsr               ;Nein: Domestic nachladen
        goto slinear           ;Ja: Linear Endlosschleife

        end
;*****

```

Digital Voltage Sensitive Relay

